INF-107 - Minicomputer Operations

Class Eleven

I. Section Three – Applications development

A. Programming Languages

B. Control Language Programs

C. RPG/400 Introduction

D. Screen Design Aid (SDA)

II. Chapter 18 - AS/400 Programming Languages

A. On-line Interactive Utilities

1. IBM Supplied

a) PDM - Programming Development Manager

(1) Front-end for maintaining libraries, objects, & files

b) SEU - Source Entry Utility

(1) Maintains source physical file members

c) DFU - Data File Utility

(1) Allows you to modify /add / delete data in database files

d) QUERY/400

(1) User friendly creation of reports

e) SDA - Screen Design Aid

(1) Allows you to “paint” a screen & then build the code to generate it.

f) RLU - Report Layout Utility

(1) Allows you to “paint” a report & then build the code to generate it.

2. Third Party Tools

a) DBU - Data Base Utility (from ProData)

(1) Like DFU, but better

b) RDA - Report Design Aid (from Gumbo Software)

(1) Like SDA, but for printer definitions

(2) Better than RLU

B. Major Programming Languages

1. Control Language (CL)

a) Used for automating operations functions

b) Used for job control

c) Built into the operating system

2. RPG/400

a) Report Program Generator

b) Invented by IBM in the 1960’s

c) Started as a 4th generation language for report generation

d) Until recently, was a fixed-format language

(1) Different fields in records have different meanings

e) Has evolved into a full function programming language

f) 95% of AS/400 applications programming has been written in RPG

JAVA

g) The most recent attempt at a universal programming language

h) Used today to add function to web pages

i) A subset of the C programming language

j) Supported at the “machine level” in the most current OS/400 releases

k) IBM is putting its money & prestige behind JAVA!

3. COBOL

a) COmmon Business Oriented Language

b) English language-like 

c) Written in Divisions, Sections, & Paragraphs

d) 1st attempt at a “universal” programming language

e) 2 varieties available on the AS/400

(1) COBOL/400

(a) ANSI (American National Standards Institute) COBOL

(2) RM/COBOL

(a) Ryan McFarland COBOL

(b) Used as an upgrade path from mini/micro computer systems

4. SQL/400

a) Structured Query Language

b) Database creation / manipulation

c) Used alone, or else in conjunction with other languages

5. C/400

a) Originally from the UNIX world

b) Also heavily used in the microcomputer world

c) A lower level language than the others

III. Chapter 19 - Control Language Programming

A. Why do Command Language programming???

1. To string together command language statements

2. To perform standard operations functions in a consistent way

3. Some system functions can only be done from a CL Program

B. PGM Command

1. The  first statement in a program

2. Allows program parameters to be defined

a) Parameter is a variable which is either input into a program or output from the program (or both)

C. DCL Command

1. Declare a variable

2. Variable is a piece of data within a program

3. Types of variables

a) Character

b) Decimal

c) Logical

Processing Commands

4. You can use either IBM supplied or user written commands

5. These are the same commands which you have used in lab (from Chapter 3)

6. By stringing the commands together, you can automate many routine operations procedures.

D. Program Control Commands

1. CALL program

a) runs the program & then returns to the program

b) You must specify any parameters needed to run the program

2. RETURN

a) Ends program

3. TFRCTL

a) Ends program & starts another program

4. ENDPGM

a) The last statement in a program

E. Logic Control Commands

1. Conditional logic (change the “flow” of program logic)

a) IF (logical expression) 

b) THEN (do this if true)

c) ELSE (do this if false)

2. Execute multiple commands as a “block”

a) DO

b)       (list of commands)

c) ENDDO

3. Change sequence of execution

a) label:

(1) A place in a program

b) GOTO   label

(1) Go to the place marked by the label

F. Comment Lines

1. Allow documentation to be entered

2. Begins with /*   and ends with   */

IV. Sample CL Programs

A. CRTSTU – Create Student User Environment

0001.00              PGM        PARM(&STUDENT)                                  

0002.00 /*                                                                  */  

0003.00 /* CRTSTU - CREATE A STUDENT USER ENVIRONMENT                       */  

0004.00 /*                                                                  */  

0005.00              DCL        VAR(&STUDENT) TYPE(*CHAR) LEN(10)               

0006.00              DCL        VAR(&JOB    ) TYPE(*CHAR) LEN(10)               

0007.00                                                                         

0008.00 /* IGNORE ERRORS THAT OCCUR                                        */   

0010.00              MONMSG     MSGID(CPF0000)                                  

0011.00                                                                         

0012.00 /* LOOK UP THE TERMINAL THAT IS RUNNING THIS JOB                   */   

0014.00              RTVJOBA    JOB(&JOB)                                       

0015.00                                                                         

0016.00 /*    CREATE AN INDIVIDUAL LIBRARY FOR THE STUDENT'S WORK          */   

0018.00              CRTLIB     LIB(&STUDENT) TEXT('IT107 Student' *BCAT +      

0019.00                           &STUDENT) AUT(*EXCLUDE) CRTAUT(*EXCLUDE)      

0020.00                 MONMSG     MSGID(CPF2111) EXEC(DO)                      

0021.00                 SNDPGMMSG  MSG('Student library' *BCAT &STUDENT *BCAT + 

0022.00                              'already exists - create student ended +   

0023.00                              abnormally')                               

0024.00                 GOTO       CMDLBL(END)                                  

0025.00                 ENDDO                                                   

0026.00                                                                         

0027.00 /*    CREATE THE STUDENT'S USER PROFILE                             */  

0029.00              CRTUSRPRF  USRPRF(&STUDENT) PASSWORD(&STUDENT) +           

0030.00                           PWDEXP(*YES) USRCLS(*PGMR) +                  

0031.00                           CURLIB(&STUDENT) INLPGM(JMYERS/SIGNON) +      

0032.00                           TEXT('IT107 Student -' *BCAT &STUDENT) +      

0033.00                           DSPSGNINF(*YES)                               

0034.00                                                                         

0035.00 /*    CREATE THE ENVIRONMENT FOR THE STUDENT'S JOBS TO RUN         */   

0037.00              CRTJOBD    JOBD(&STUDENT/&STUDENT) USER(&STUDENT) +        

0038.00                           PRTTXT('Output for ' *CAT &STUDENT) +         

0039.00                           RTGDTA(QCMDB) INLLIBL(QTEMP &STUDENT +        

0040.00                           JMYERSDATA QGPL QRPG)                         

0041.00              CHGUSRPRF  USRPRF(&STUDENT) JOBD(&STUDENT/&STUDENT)        

0042.00                                                                         

0043.00 /*    CREATE SOURCE PHYSICAL FILES TO HOLD THE STUDENT'S WORK      */   

0045.00              CRTSRCPF   FILE(&STUDENT/QDDSSRC) TEXT('DDS Source Code')  

0046.00              CRTSRCPF   FILE(&STUDENT/QCLSRC ) TEXT('CL  Source Code')  

0047.00              CRTSRCPF   FILE(&STUDENT/QCMDSRC) TEXT('CMD Source Code')  

0048.00              CRTSRCPF   FILE(&STUDENT/QRPGSRC) TEXT('RPG Source Code')  

0049.00              CRTSRCPF   FILE(&STUDENT/QCBLSRC) TEXT('COBOL Source +     

0050.00                           Code')                                        

0051.00              CRTSRCPF   FILE(&STUDENT/QTXTSRC) TEXT('Text file')        

0053.00                                                                         

0054.00 /*    EXCLUDE ALL ACCESS TO THE STUDENT'S LIBRARY                  */   

0055.00 /*       EXCEPT BY THE STUDENT                                     */   

0056.00                                                                         

0057.00              RVKOBJAUT  OBJ(QSYS/&STUDENT) OBJTYPE(*LIB) USER(*ALL) +   

0058.00                           AUT(*ALL)                                     

0059.00              GRTOBJAUT  OBJ(QSYS/&STUDENT) OBJTYPE(*LIB) +              

0060.00                           USER(&STUDENT) AUT(*ALL)                      

0061.00              GRTOBJAUT  OBJ(&STUDENT/*ALL) OBJTYPE(*ALL) +              

0062.00                           USER(&STUDENT) AUT(*ALL)                      

0063.00              CHGOBJOWN  OBJ(QSYS/&STUDENT) OBJTYPE(*LIB) +              

0064.00                           NEWOWN(&STUDENT) CUROWNAUT(*SAME)             

0065.00                                                                         

0066.00 /*    NOTIFY USER OF SUCCESSFUL CONCLUSION                         */   

0068.00              SNDPGMMSG  MSG('Student environment' *BCAT &STUDENT +      

0069.00                           *BCAT 'created successfully')                 

0071.00  END:        ENDPGM                                                     

SIGNON – Standard Student Sign-on Program

0001.00              PGM                                                        

0002.00                                                                         

0003.00 /*  SIGNON - STANDARD STUDENT SIGN ON PROGRAM                        */ 

0004.00                                                                         

0005.00              DCL        VAR(&TERMINAL  ) TYPE(*CHAR) LEN(10)            

0006.00              DCL        VAR(&STUDENT   ) TYPE(*CHAR) LEN(10)            

0007.00                                                                         

0008.00 /*  IGNORE ERRORS UNLESS OTHERWISE SPECIFIED                         */ 

0009.00                                                                         

0010.00              MONMSG     MSGID(CPF0000)                                  

0011.00                                                                         

0012.00 /*  GET THE TERMINAL & STUDENT NAMES                                 */ 

0013.00                                                                         

0014.00              RTVJOBA    JOB(&TERMINAL) USER(&STUDENT)                   

0015.00                                                                         

0016.00 /*  SET THE TERMINAL MESSAGE QUEUE INTO BREAK MODE                   */

0017.00                                                                         

0018.00              CHGMSGQ    MSGQ(&TERMINAL) DLVRY(*BREAK) SEV(00)           

0019.00              MONMSG     MSGID(CPF2451) EXEC(SNDPGMMSG MSG('Your +       

0020.00                           userid is already signed on.')) /* Userid +   

0021.00                           already logged on */                          

0022.00                                                                         

0023.00 /*  SET UP THE USER'S LIBRARY LIST                                   */ 

0024.00                                                                         

0025.00              CHGLIBL    LIBL(QTEMP &STUDENT JMYERSDATA QGPL QRPG)       

0026.00              CHGCURLIB  CURLIB(&STUDENT)                                

0027.00                                                                         

0028.00 /*  DISPLAY THE SYSTEM MAIN MENU                                     */ 

0029.00                                                                         

0030.00              GO         MENU(MAIN) RTNPNT(*NO)                          

0031.00                                                                         

0032.00                                                                         

0033.00              ENDPGM

B. BACKUP – Nightly Backup of User Objects 

0001.00              PGM        PARM(&PWRDWN &IPLDATTIM &DLY)                   

0002.00                                                                         

0003.00 /*                                                                  */  

0004.00 /*  BACKUP - PERFORM DAILY SYSTEM BACKUP TO TAPE                    */  

0005.00 /*           (THIS IS THE COMMAND PROCESSING PROGRAM                */  

0006.00 /*             FOR USER COMMAND BACKUP IN LIBRARY BACKUP)           */  

0007.00 /*           PARAMETERS:                                            */  

0008.00 /*             &PWRDWN - POWER DOWN THE SYSTEM WHEN DONE?           */  

0009.00 /*                        (VALUES ARE "Y" OR "N")                   */  

0010.00 /*             &IPLDATTIM - WHEN TO POWER SYSTEM UP?                */  

0011.00 /*             &DLY       - TIME TO DELAY START OF OPERATION TILL   */  

0012.00 /*                                                                  */  

0013.00                                                                         

0014.00              DCL        VAR(&PWRDWN) TYPE(*CHAR) LEN(1)                 

0015.00              DCL        VAR(&IPLDATTIM) TYPE(*CHAR) LEN(20)             

0016.00              DCL        VAR(&DLY) TYPE(*CHAR) LEN(6)

0017.00              DCL        VAR(&COSBS) TYPE(*CHAR) LEN(160) +              

0018.00                           VALUE('Copyright, Strategic Business +        

0019.00                           Systems - 1986, 1987   All Rights Reserved')  

0020.00              DCL        VAR(&MSG) TYPE(*CHAR) LEN(80)                   

0021.00                                                                         

0022.00              RMVMSG     CLEAR(*ALL)                                     

0023.00                                                                         

0024.00              OVRPRTF    FILE(QPJOBLOG) OUTQ(BACKUP/BACKUP)              

0025.00                                                                         

0026.00              IF         COND(&IPLDATTIM ¬= '*SAME') THEN(DO)            

0027.00                  CHGSYSVAL SYSVAL(QIPLDATTIM) VALUE(&IPLDATTIM)         

0028.00                  MONMSG MSGID(CPF0000)                                  

0029.00                  ENDDO                                                  

0030.00                                                                         

0031.00              DLYJOB     RSMTIME(&DLY)                                   

0032.00                                                                         

0033.00 SAVCMDS:

0034.00 /*                                                                  */ 

0035.00 /*   SAVE ALL USER LIBRARIES                                        */ 

0036.00 /*                                                                  */ 

0037.00              SAVLIB     LIB(*ALLUSR) DEV(TAP02) SEQNBR(1) +            

0038.00                           EXPDATE(010193) ENDOPT(*LEAVE) +             

0039.00                           OMITLIB(BACKUP)                              

0040.00                                                                        

0041.00              MONMSG     MSGID(CPF3777 CPF4024)                         

0042.00 /*                                                                  */ 

0043.00 /*    SAVE DOCUMENT LIBRARY OBJECTS (FOLDERS)                       */ 

0044.00 /*                                                                  */ 

0045.00              SAVDLO     DLO(*ALL) DEV(TAP02) EXPDATE(010193) +         

0046.00                           ENDOPT(*LEAVE)                               

0047.00                                                                        

0048.00              MONMSG     MSGID(CPI9025 CPF4024)                         

0049.00                                                                        

0050.00 /*                                                                  */ 

0051.00 /*     SAVE INTEGRATED FILE SYSTEM                                  */ 

0052.00 /*                                                                  */ 

0053.00              SAV        DEV('QSYS.LIB/TAP02.DEVD') OBJ(('/*' +         

0054.00                           *INCLUDE) ('/QSYS.LIB' *OMIT) ('/QDLS' +     

0055.00                           *OMIT)) EXPDATE(010193) ENDOPT(*UNLOAD) +    

0056.00                           UPDHST(*YES)                                 

0057.00                                                                        

0058.00              MONMSG     MSGID(CPI3723 CPF4024 CPF3837 CPA4036)         

0059.00 /*                                                                  */ 

0060.00 /*   IF THE SYSTEM IS TO BE POWERED DOWN,                           */ 

0061.00 /*      SUBMIT THE COMMAND TO POWER DOWN THE SYSTEM                 */ 

0062.00 /*                                                                  */ 

0063.00 PWRDOWN:                                                               

0064.00                                                                        

0065.00              IF         COND(&PWRDWN = 'Y') THEN(DO)                   

0066.00              SBMJOB     CMD(PWRDWNSYS DELAY(60)) JOB(PWRDWNSYS)        

0067.00              ENDDO

0068.00                                                                         

0069.00 /*                                                                  */  

0070.00 /*  MESSAGE HANDLING ROUTINE                                        */  

0071.00 /*                                                                  */  

0072.00 CPFMSG:                                                                 

0073.00                                                                         

0074.00              RCVMSG     RMV(*NO) MSG(&MSG)                              

0075.00                                                                         

0076.00              IF         COND(&MSG ¬= ' ') THEN(DO)                      

0077.00              SNDPGMMSG  MSG(&MSG) TOPGMQ(*PRV)                          

0078.00              GOTO       CMDLBL(CPFMSG)                                  

0079.00              ENDDO                                                      

0080.00                                                                         

0081.00              ENDPGM

V. Creating CL Programs

A. Use the source entry utility to enter the source program.  The easiest way is to use PDM)

1. Source physical file is QCLSRC in your library

2. Member type is “CLP” (Command Language Program)

B. Use the CRTCLPGM command (option 14 in PDM) to create the program

1. If you get a message back that the job finished “normally”, the program has been created

2. If your job finished “abnormally”, you will need to look at the output in order to determine what went wrong.  Here are two ways to look at the output.

a) Use the WRKSBMJOB command.  Using option 8 (Work with spool files) in front of your last submitted job will show you the output for the job.  Placing option 5 in front of a spool file will allow you to display it on the screen.

b) Within the Source Entry Utility, you can use function key F15 in order to look at printouts for the source member which you are editing.  Select option 2 (Spool file) and press enter.  The last spool file generated for this file will be displayed on the bottom hald of thescreen.  In the SEU(_____ command line (for the bottom half of the screen), you can enter F ‘*’ 1 (Find a * in column 1 of the output).  This will position you to the first error message in the printout (you can fix the error in the top window simultaneously).  Pressing F16 will find the next error message.

VI. Lab Exercise

A. Enter the following Command Language Program into member LAB11 type CLP of file QCLSRC in your library:


0001.00              PGM        PARM(&TO)                                       

0002.00 /*                                                                   */ 

0003.00 /*  LAB11 - GENERATE AND SEND A MESSAGE                              */ 

0004.00 /*                                                                   */ 

0005.00                                                                         

0006.00              DCL        VAR(&TO ) TYPE(*CHAR) LEN(10)                   

0007.00                                                                         

0008.00              MONMSG     MSGID(CPF0000) EXEC(GOTO CMDLBL(ERROR))         

0009.00                                                                         

0010.00              SNDMSG     MSG('I sent a user message under program +      

0011.00                           control!') TOUSR(&TO)                         

0012.00                                                                         

0013.00              SNDPGMMSG  MSG('The message was successfully sent')        

0014.00                                                                         

0015.00              GOTO       CMDLBL(END)                                     

0016.00

0017.00 /*                                                                   */ 

0018.00 /*   ERROR HANDLING ROUTINE                                          */ 

0019.00 /*                                                                   */ 

0020.00  ERROR:                                                                 

0021.00              SNDPGMMSG  MSG('The message was not sent due to errors')   

0022.00                                                                         

0023.00  END:        ENDPGM

B. Compile the program.  If you cannot solve the compile time errors, contact the instructor.

C. Execute the program …  CALL  LAB11  JMYERS
 the instructor should get a message.

D. Use the screen print function to print the following:

1. The source code of your program

2. The first page of the output of the command DSPPGM   LAB11

E. Staple your work together & hand it in to the instructor.

